

气缸是工业现场使用较多的一种执行器，它使用压缩空气作为动力，有两个进气孔，通过控制进气与排气，推动气缸往复运动。一些安全系数较高的气缸还配有抱闸。默认情况下抱闸闭合，气缸不能运动。在通入压缩空气后抱闸打开，气缸可自由运动。今天这篇文章我们来编写一个用于控制气缸往复运动的函数块——FB5022_Cylinder，它也可用于液压缸的类似场合。

SIEMENS
Ingenuity for life

西门子PLC高级编程语言SCL实例讲解

——气缸控制函数块

<https://www.founderchip.com>

函数块FB5022_Cylinder有如下一些特点：

- 支持自动模式和维护（手动）模式。在自动模式下，气缸会运动某个指定的时间，直到时间到达或者触发前进/后退到位反馈信号；在维护（手动）模式下，气缸运动跟随按钮信号，按下按钮运动，松开按钮停止；
- 气缸运动时会使能抱闸松开信号；
- 当同时触发前进/后退请求或者前进/后退的到位信号时，会报告错误；

下面我们来编写函数。

打开博途开发环境，新建函数块FB5022_Cylinder，添加变量如下图所示：

FB5022_Cylinder						
	名称	数据类型	默认值	保持	设定值	注释
1	▼ Input					
2	release	Bool	false	非保持	<input type="checkbox"/>	使能气缸运动(1=ok)
3	reqForward	Bool	false	非保持	<input type="checkbox"/>	自动模式下请求前进
4	reqBackward	Bool	false	非保持	<input type="checkbox"/>	自动模式下请求后退
5	btnForward	Bool	false	非保持	<input type="checkbox"/>	维护/手动模式下请求前进
6	btnBackward	Bool	false	非保持	<input type="checkbox"/>	维护/手动模式下请求后退
7	timeForward	Time	T#0ms	非保持	<input type="checkbox"/>	前进的时间
8	timeBackward	Time	T#0ms	非保持	<input type="checkbox"/>	后退的时间
9	mainMode	Bool	false	非保持	<input type="checkbox"/>	维护模式
10	feedbackForward	Bool	false	非保持	<input type="checkbox"/>	前进到位信号
11	feedbackBackward	Bool	false	非保持	<input type="checkbox"/>	后退到位信号
12	▼ Output					
13	forward	Bool	false	非保持	<input type="checkbox"/>	前进
14	backward	Bool	false	非保持	<input type="checkbox"/>	后退
15	brake	Bool	false	非保持	<input type="checkbox"/>	抱闸 (1=打开)
16	error	Bool	false	非保持	<input type="checkbox"/>	1=错误
17	▼ InOut					
18	<新增>					
19	▼ Static					
20	IEC_Timer_TP_Forward	TP_TIME		非保持	<input checked="" type="checkbox"/>	前进脉冲定时器
21	IEC_Timer_TP_Backwa...	TP_TIME		非保持	<input checked="" type="checkbox"/>	后退脉冲定时器
22	statTimerForwardQ	Bool	false	非保持	<input checked="" type="checkbox"/>	前进激活
23	statTimerBackwardQ	Bool	false	非保持	<input type="checkbox"/>	后退激活
24	statForward	Bool	false	非保持	<input type="checkbox"/>	前进
25	statBackward	Bool	false	非保持	<input type="checkbox"/>	后退
26	statError	Bool	false	非保持	<input type="checkbox"/>	错误状态
27	statForwardMaint	Bool	false	非保持	<input type="checkbox"/>	维护/手动模式下前进
28	statBackwardMaint	Bool	false	非保持	<input type="checkbox"/>	维护/手动模式下后退
29	▼ Temp					
30	tmpReqForward	Bool			<input type="checkbox"/>	前进
31	tmpReqBackward	Bool			<input type="checkbox"/>	后退
32	▼ Constant					
33	<新增>					

其中:

①输入参数:

- release: 布尔型, 1=使能函数块;
- reqForward: 布尔型, 在自动模式下请求前进;
- reqBackward: 布尔型, 在自动模式请求后退;
- btnForward: 布尔型, 维护(手动)模式下请求前进;
- btnBackward: 布尔型, 维护(手动)模式下请求后退;
- timeForward: 时间型, 自动模式下前进的最长时间;
- timeBackward: 时间型, 自动模式下后退的最长时;
- mainMode: 布尔型, 1=维护/手动模式;
- feedbackForward: 布尔型, 前进到位反馈信号;
- feedbackBackward: 布尔型, 后退到位反馈信号;

②输出参数:

- forward: 布尔型, 前进;
- backward: 布尔型, 后退;
- brake: 布尔型, 抱闸输出控制 (1=松开);

- error: 布尔型, 错误 (1=有错误)

③静态变量和临时变量:

- IEC_Timer_TP_Forward和IEC_Timer_TP_Backward是添加定时器指令时自动生成的多重背景数据块变量;
- 其它变量请看图中解释;

为了便于阅读, 我将代码分成了三个区域: 自动、维护 (手动) 和控制, 请看下面的图片:

IF...	CASE... OF...	FOR... TO DO...	WHILE... DO...	(*...*)	REGION
1					(*
2					Copyrights @Founderchip
3					=====
4					功能说明:
5					该程序实现气缸的往复运动控制, 支持抱闸控制 (1=打开抱闸)
6					输入参数:
7					release : 使能函数块
8					reqForward : 请求前进 (自动模式下)
9					reqBackward : 请求后退 (自动模式下)
10					btnForward : 请求前进 (维护/手动模式下)
11					btnBackward : 请求后退 (维护/手动模式下)
12					timeForward : 前进的时间
13					timeBackward : 后退的时间
14					mainMode : 维护/手动模式
15					feedbackForward : 前进到位反馈
16					feedbackBackward : 后退到位反馈
17					
18					输出参数:
19					forward : 前进
20					backward : 后退
21					brake : 抱闸 (1=松开)
22					error : 错误 (1=有错误)
23					
24					
25					作者:
26					北岛李工
27					2023-6-27
28					-----
29					修改日志:
30					
31					2023-6-27 v1.0 版本(首发) 北岛李工
32					
33					=====
34					*)

```

35 REGION autoMode
36 // 自动模式（定时器到达时间或遇到前进到位信号停止）
37 //自动模式下向前运动命令
38 IF NOT #mainMode AND #reqForward AND NOT #reqBackward THEN
39     #tmpReqForward := TRUE;
40     RESET_TIMER(#IEC_Timer_TP_Backward);
41 END_IF;
42 //脉冲定时器
43 #IEC_Timer_TP_Forward(IN := #tmpReqForward,
44                       PT := #timeForward,
45                       Q => #statTimerForward);
46 //自动模式下向后运动命令
47 IF NOT #mainMode AND #reqBackward AND NOT #reqForward THEN
48     #tmpReqBackward := TRUE;
49     RESET_TIMER(#IEC_Timer_TP_Forward);
50 END_IF;
51 //脉冲定时器
52 #IEC_Timer_TP_Backward(IN := #tmpReqBackward,
53                        PT := #timeBackward,
54                        Q => #statTimerBackward);
55 END_REGION
56

```

```

57 REGION maintMode
58 //维护模式
59 //手动模式下前进命令
60 IF #mainMode AND #btnForward AND NOT #btnBackward THEN
61     #statForwardMaint := TRUE;
62 ELSE
63     #statForwardMaint := FALSE;
64 END_IF;
65 //手动模式下后退命令
66 IF #mainMode AND #btnBackward AND NOT #btnForward THEN
67     #statBackwardMaint := TRUE;
68 ELSE
69     #statBackwardMaint := FALSE;
70 END_IF;
71 END_REGION
72

```

```

73 REGION control
74     //具体控制代码
75     //错误状态
76 IF (#feedbackBackward AND #feedbackForward)
77     OR (#reqForward AND #reqBackward)
78     OR (#btnForward AND #btnBackward) THEN
79     #statError := TRUE;
80 ELSE
81     #statError := FALSE;
82 END_IF;
83
84 //前进
85 IF #release AND NOT #statError AND NOT #feedbackForward AND
86     (#statTimerForwardQ OR #statForwardMaint) THEN
87     #statForward := TRUE;
88 ELSE
89     #statForward := FALSE;
90 END_IF;
91 //后退
92 IF #release AND NOT #statError AND NOT #feedbackBackward AND
93     (#statTimerBackwardQ OR #statBackwardMaint) THEN
94     #statBackward := TRUE;
95 ELSE
96     #statBackward := FALSE;
97 END_IF;
98 //抱闸
99 IF #statForward OR #statBackward THEN
100     #brake := TRUE;
101 ELSE
102     #brake := FALSE;
103 END_IF;
104 //输出
105 #forward := #statForward;
106 #backward := #statBackward;
107 #error := #statError;
108 END_REGION
109

```


在OB1中调用该函数块进行测试，如下图所示：

▼ 块标题： "Main Program Sweep (Cycle)"

注释

▼ 程序段 1： test

注释




如果你有任何问题，欢迎留言。

下面是西门子SCL编程的文章归档链接：

[》》 西门子SCL编程入门到精通文章归档 《《](#)

我的书《西门子S7-1200/1500 PLC SCL语言编程——从入门到精通》从硬件到软件，比较详细的介绍了SCL语言的编程，感兴趣的话可以扫描下面的二维码查看：



识别图中小
程序码购买